



```
In[1]:= (* potential *)
(* ----- *)
v[x_] := -1 / (1 + x^2)
Plot[v[x], {x, -5, 5}]
```



Out[2]= - Graphics -

```
In[3]:= (* plot solution of schroedinger equation for trial e *)
(* note: all numerical constants (hbar,m,..)=1 *)
(* ----- *)
plotsol[e_] :=
Block[{x0 = 10}, sol = NDSolve[{-psi''[x] + (v[x] - e) * psi[x] == 0, psi[-x0] == 0.001,
psi'[-x0] == 0.001}, psi, {x, -x0, x0}];
Plot[Evaluate[psi[x] /. sol], {x, -x0, x0}]]
```


```
In[4]:= (* you can play with this ! *)
(* ----- *)
plotsol[-0.437]
```



Out[4]= - Graphics -

```
In[5]:= (* shooting method *)
(* ----- *)
shoot[e_] :=
Block[{x0 = 10}, sol = NDSolve[{-psi''[x] + (v[x] - e) * psi[x] == 0, psi[-x0] == 0.001,
psi'[-x0] == 0.001}, psi, {x, -x0, x0}];
psi[
x0] /.
sol]
```


```
In[6]:= Plot[shoot[e], {e, -0.6, 0}, PlotRange -> {-10, 10}]
```



```
Out[6]= - Graphics -
```

```
In[7]:= (* find ground state energy *)
(* ----- *)
Off[NDSolve::ndnum]
Off[NDSolve::dsvar]
Off[ReplaceAll::reps]
e0 = e /. FindRoot[shoot[e] == 0, {e, -0.5}]
plotsol[e0]
```

```
Out[10]= -0.437881
```




```
Out[11]= - Graphics -
```

```

In[12]:= (* look at solutions for e0 +/- de *)
(* ----- *)
x0 = 10;
e = e0; sol0 = NDSolve[{-psi''[x] + (v[x] - e) * psi[x] == 0,
  psi[-x0] == 0.001, psi'[-x0] == 0.001}, psi, {x, -x0, x0}];
e = e0 + 0.01;
sol1 = NDSolve[{-psi''[x] + (v[x] - e) * psi[x] == 0,
  psi[-x0] == 0.001, psi'[-x0] == 0.001}, psi, {x, -x0, x0}];
e = e0 - 0.01;
sol2 = NDSolve[{-psi''[x] + (v[x] - e) * psi[x] == 0,
  psi[-x0] == 0.001, psi'[-x0] == 0.001}, psi, {x, -x0, x0}];
Plot[{Evaluate[psi[x] /. sol1], Evaluate[psi[x] /. sol2], Evaluate[psi[x] /. sol0]},
  {x, -x0, x0}, PlotStyle -> {RGBColor[1, 0, 0], RGBColor[0, 0, 1], Thickness[0.01]},
  PlotRange -> {-0.3, 0.5}]

```



Out[18]= - Graphics -